
*************************************************************************
*** FUNIMP.PRG.: Modulo que tiene las funciones especiales para la    ***
***              impresora. Codigos de Control especiales.            ***
*** Fecha.: Octubre de 1995                                           ***
*** Autor.: Ana I Almaden para J.A.S. Software                        ***
*** Reservados todos los derechos sobre el presente programa          ***
*************************************************************************


************************ IniciaIMPRE ************************************
***
*** Funcion que inicializa la impresora.
*** En este momento solo le manda el largo del papel en lineas.
***
*** PARAMETROS:
***
***      nlarlin => largo del papel en lineas (actualmente 66)
***
*************************************************************************

function iniciaIMPRE(nlarlin)
LOCAL fil, col, impre

impre := SET(_SET_PRINTER,.T.)

fil := PROW()
col := PCOL()


@01,01 SAY CHR(27)+CHR(67)+chr(nlarlin)

SETPRC (fil,col)
SET(_SET_PRINTER,impre)

return NIL


*************************** Control ************************************
***
*** Funcion que manda un codigo de control a la impresora.
*** Esta funcion es necesaria porque los contadores de fila y columna
*** que lleva la impresora cambian cuando se manda un codigo de control.
*** Entonces los contadores que llevamos nosotros van descompensados.
*** Lo que hace esta funcion es no cambiar los contadores de la impresora.
***
*** PARAMETROS:
***
***      cod => codigo de control. Puede ser numerico o de caracter.
***
*************************************************************************

function CONTROL(cod)
LOCAL cCOD
LOCAL fil, col, impre

if EMPTY(cod)
   return ""
endif

if VALTYPE (cod) == "N"
   cCOD := CHR(cod)
else
   cCOD := cod
endif

impre := SET(_SET_PRINTER,.T.)

fil := PROW()
col := PCOL()

@fil,col say cCOD

SETPRC (fil,col)
SET(_SET_PRINTER,impre)

return ""

